Math 277
Ordinary Differential Equations

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 9

The Method of Undetermined Coefficients

Initializations

```
> restart;
```

9.1 The Method of Undetermined Coefficients

This method seeks to find a particular solution of the differential equation

$$ay'' + b\,y' + cy = f(t)$$

in case that the function $f(t)$ has a simple differential family. The examples below explore various choices of $f(t)$. Mathematical details will be provided in class.

Examples

Example 9.1.1

Find a particular solution of the differential equation

$$2\,y'' + 5\,y' - y = t^3$$

Solution

We search for a particular solution of the form

$$y_p(t) = A_0 + A_1\,t + A_2\,t^2 + A_3\,t^3$$

Code the differential equation and the trial solution.

```
> deq:=2*diff(y(t), t$2)+5*diff(y(t), t)-y(t)=t^3;
  ytry:=add(A[k]*t^k, k=0..3);
```

$$deq := 2\left(\frac{d^2}{dt^2}\,y(t)\right) + 5\left(\frac{d}{dt}\,y(t)\right) - y(t) = t^3$$

$$ytry := A_0 + A_1\,t + A_2\,t^2 + A_3\,t^3 \qquad \textbf{(2.1.1.1)}$$

Substitute the trial solution into the differential equation and compare the coefficients of t^n, $n = 0, 1, 2, 3$.

```
> eq:=eval(deq, y(t)=ytry);
```

$$eq := 4\,A_2 + 12\,A_3\,t + 5\,A_1 + 10\,A_2\,t + 15\,A_3\,t^2 - A_0 - A_1\,t - A_2\,t^2 - A_3\,t^3 = t^3 \qquad \textbf{(2.1.1.2)}$$

```
> pars:=solve(identity(eq, t), {seq(A[k], k=0..3)});
```

$$pars := \{A_0 = -870, A_1 = -162, A_2 = -15, A_3 = -1\} \qquad \textbf{(2.1.1.3)}$$

The particular solution is given by

```
> yp:=eval(ytry, pars);
```

$$yp := -870 - 162\,t - 15\,t^2 - t^3 \quad (2.1.1.4)$$

Example 9.1.2

Find a particular solution of the differential equation

$$y'' + y' - 6\,y = te^{5\,t}$$

Solution

First determine the roots of the auxiliary equation of the corresponding homogeneous equation.

```
> aux:=r^2+r-6=0;
  ev:=solve(aux, r);
```

$$aux := r^2 + r - 6 = 0$$

$$ev := 2,\ -3 \quad (2.1.2.1)$$

Since 5 is not a root of the auxiliary equation, we search for a particular solution of the form

$$y_p(t) = \left(A_0 + A_1\,t\right)\,e^{5\,t}$$

Code the differential equation and the trial solution.

```
> deq:=diff(y(t), t$2)+diff(y(t), t)-6*y(t)=t*exp(5*t);
  ytry:=add(A[k]*t^k, k=0..1)*exp(5*t);
```

$$deq := \frac{d^2}{dt^2}\,y(t) + \frac{d}{dt}\,y(t) - 6\,y(t) = t\,e^{5\,t}$$

$$ytry := \left(A_0 + A_1\,t\right)\,e^{5\,t} \quad (2.1.2.2)$$

Substitute the trial solution into the differential equation and compare the coefficients of $e^{5\,t}$ and $te^{5\,t}$.

```
> eq:=eval(deq, y(t)=ytry);
```

$$eq := 11\,A_1\,e^{5\,t} + 24\,\left(A_0 + A_1\,t\right)\,e^{5\,t} = t\,e^{5\,t} \quad (2.1.2.3)$$

```
> pars:=solve(identity(eq, t), {seq(A[k], k=0..1)});
```

$$pars := \left\{A_0 = -\frac{11}{576},\ A_1 = \frac{1}{24}\right\} \quad (2.1.2.4)$$

The particular solution is given by

```
> yp:=eval(ytry, pars);
```

$$yp := \left(-\frac{11}{576} + \frac{1}{24}\,t\right)\,e^{5\,t} \quad (2.1.2.5)$$

Example 9.1.3

Find a particular solution of the differential equation

$$y'' + y' - 6\,y = te^{2\,t}$$

Solution

First determine the roots of the auxiliary equation of the corresponding homogeneous equation.

```
> aux:=r^2+r-6=0;
```

```
ev:=solve(aux, r);
```

$$aux := r^2 + r - 6 = 0$$

$$ev := 2, -3 \quad \textbf{(2.1.3.1)}$$

Since 2 is a root of the auxiliary equation, we search for a particular solution of the form

$$y_p(t) = t\left(A_0 + A_1 t\right) e^{2t}$$

Code the differential equation and the trial solution.

```
> deq:=diff(y(t), t$2)+diff(y(t), t)-6*y(t)=t*exp(2*t);
  ytry:=t*add(A[k]*t^k, k=0..1)*exp(2*t);
```

$$deq := \frac{d^2}{dt^2} y(t) + \frac{d}{dt} y(t) - 6\, y(t) = t\, e^{2t}$$

$$ytry := t\left(A_0 + A_1 t\right) e^{2t} \quad \textbf{(2.1.3.2)}$$

Substitute the trial solution into the differential equation and compare the coefficients of e^{2t} and te^{2t}.

```
> eq:=eval(deq, y(t)=ytry);
```

$$eq := 2\, A_1\, e^{2t} + 5\left(A_0 + A_1 t\right) e^{2t} + 5\, t\, A_1\, e^{2t} = t\, e^{2t} \quad \textbf{(2.1.3.3)}$$

```
> pars:=solve(identity(eq, t), {seq(A[k], k=0..1)});
```

$$pars := \left\{A_0 = -\frac{1}{25}, A_1 = \frac{1}{10}\right\} \quad \textbf{(2.1.3.4)}$$

The particular solution is given by

```
> yp:=eval(ytry, pars);
```

$$yp := t\left(-\frac{1}{25} + \frac{1}{10}\, t\right) e^{2t} \quad \textbf{(2.1.3.5)}$$

Example 9.1.4

Find a particular solution of the differential equation

$$y'' + 6\, y' + 25\, y = t^2 e^{-3t} \cos 4t$$

Solution

First determine the roots of the auxiliary equation of the corresponding homogeneous equation.

```
> aux:=r^2+6*r+25=0;
  ev:=solve(aux, r);
```

$$aux := r^2 + 6\, r + 25 = 0$$

$$ev := -3 + 4\, \mathrm{I}, -3 - 4\, \mathrm{I} \quad \textbf{(2.1.4.1)}$$

Since $-3 + 4\, i$ is a root of the auxiliary equation, we search for a particular solution of the form

$$y_p(t) = t\left(A_0 + A_1 t + A_2 t^2\right) e^{-3t} \cos(4\, t) + t\left(B_0 + B_1 t + B_2 t^2\right) e^{-3t} \sin(4\, t)$$

Code the differential equation and the trial solution.

```
> deq:=diff(y(t), t$2)+6*diff(y(t), t)+25*y(t)=t^2*exp(-3*
  t)*cos(4*t);
```

```
ytry:=t*add(A[k]*t^k, k=0..2)*exp(-3*t)*cos(4*t)+t*add(B[k]*t^k, k=0..2)*exp(-3*t)*sin(4*t);
```

$$deq := \frac{d^2}{dt^2}y(t) + 6\left(\frac{d}{dt}y(t)\right) + 25\,y(t) = t^2 e^{-3t}\cos(4\,t)$$

$$ytry := t\left(A_0 + A_1\,t + A_2\,t^2\right) e^{-3t}\cos(4\,t) + t\left(B_0 + B_1\,t + B_2\,t^2\right) e^{-3t}\sin(4\,t) \quad \textbf{(2.1.4.2)}$$

Substitute the trial solution into the differential equation and compare the coefficients of similar terms

```
> eq:=eval(deq, y(t)=ytry);
```

$$\begin{aligned} eq := {} & 2\left(A_1 + 2\,A_2\,t\right) e^{-3t}\cos(4\,t) - 8\left(A_0 + A_1\,t + A_2\,t^2\right) e^{-3t}\sin(4\,t) \\ & + 2\left(B_1 + 2\,B_2\,t\right) e^{-3t}\sin(4\,t) + 8\left(B_0 + B_1\,t + B_2\,t^2\right) e^{-3t}\cos(4\,t) \\ & + 2\,t\,A_2\,e^{-3t}\cos(4\,t) - 8\,t\left(A_1 + 2\,A_2\,t\right) e^{-3t}\sin(4\,t) \\ & + 2\,t\,B_2\,e^{-3t}\sin(4\,t) + 8\,t\left(B_1 + 2\,B_2\,t\right) e^{-3t}\cos(4\,t) = t^2 e^{-3t}\cos(4\,t) \end{aligned} \quad \textbf{(2.1.4.3)}$$

```
> pars:=solve(identity(eq, t), {seq(A[k], k=0..2), seq(B[k], k=0..2)});
```

$$pars := \left\{A_0 = 0,\ A_1 = \frac{1}{64},\ A_2 = 0,\ B_0 = -\frac{1}{256},\ B_1 = 0,\ B_2 = \frac{1}{24}\right\} \quad \textbf{(2.1.4.4)}$$

The particular solution is given by

```
> yp:=eval(ytry, pars);
```

$$yp := \frac{1}{64}\,t^2 e^{-3t}\cos(4\,t) + t\left(-\frac{1}{256} + \frac{1}{24}\,t^2\right) e^{-3t}\sin(4\,t) \quad \textbf{(2.1.4.5)}$$